Clog Doc: rate1.4.tex

Temperature Equilibration Rate in Clog

Robert Singleton

Research Notes

Project:
Clog Doc
Path of TeX Source:
Clog/doc/dedx/rate1.4.tex
Last Modified By:
Robert Singleton
3/4/2020
Date Started:
3/4/2020
Date:
v1.4 Thursday 5th March, 2020 06:41

Temperature Equilibration Rate in Clog

Robert L Singleton Jr
School of Mathematics
University of Leeds
LS2 9JT

(Dated: 3/4/2020)

Abstract

Physics documentation for the BPS temperature equilibration in the code Clog.

Contents

I. GENERAL ANALYTIC EXPRESSIONS FOR THE BPS RATE COEFFICIENTS

Suppose we have a plasma with various species labeled by an index b at distinct temperatures T_b, number densities n_b, and species masses m_b.[1] Temperature will be measured in energy units, and we denote the inverse temperature by $\beta_b = 1/T_b$. Electrostatic units will be rationalized cgs. Species a and b will exchange energy through coulomb interactions, and the rate of change in the energy density between species a and b at temperatures T_a and T_b is given by the usual rate equation

$$\frac{d\mathcal{E}_{ab}}{dt} = -\,\mathcal{C}_{ab}\,\Big(T_a - T_b\Big)\ . \tag{1.1}$$

The rate coefficients $\mathcal{C}_{ab}$ as calculated by BPS take the form

$$\mathcal{C}_{ab} = \Big(\mathcal{C}^{\rm C}_{ab,{\rm R}} + \mathcal{C}^{\rm C}_{ab,{\rm S}}\Big) + \mathcal{C}^{\rm QM}_{ab}\ , \tag{1.2}$$

where the first two term are purely classical and are given by a long- and a short-distance contribution,

$$\mathcal{C}^{\rm C}_{ab,{\rm R}} \ = \ \frac{\kappa_a^2\,\kappa_b^2}{2\pi}\left(\frac{\beta_a m_a}{2\pi}\right)^{1/2}\left(\frac{\beta_b m_b}{2\pi}\right)^{1/2}\int_{-\infty}^{\infty} dv\, v^2 e^{-\frac{1}{2}(\beta_a m_a+\beta_b m_b)v^2}\,\frac{i}{2\pi}\,\frac{F(v)}{\rho_{\rm tot}(v)}\ln\!\left\{\frac{F(v)}{\kappa_e^2}\right\} \tag{1.3}$$

$$\mathcal{C}^{\rm C}_{ab,{\rm S}} \ = -\kappa_a^2\,\kappa_b^2\,\frac{(\beta_a m_a\beta_b m_b)^{1/2}}{(\beta_a m_a+\beta_b m_b)^{3/2}}\,\left(\frac{1}{2\pi}\right)^{3/2}\,\left[\ln\!\left\{\frac{e_a\,e_b}{4\pi}\,\frac{\kappa_e}{4\,m_{ab}\,V_{ab}^2}\right\}+2\gamma\right] \tag{1.4}$$

respectively, and the third term is the short-distance quantum scattering piece (this term vanishes as $\hbar\to 0$)

$$\mathcal{C}^{\rm QM}_{ab} = -\frac{1}{2}\,\kappa_a^2\,\kappa_b^2\,\frac{(\beta_a m_a\,\beta_b m_b)^{1/2}}{(\beta_a m_a+\beta_b m_b)^{3/2}}\left(\frac{1}{2\pi}\right)^{3/2}\int_0^\infty d\zeta\, e^{-\zeta/2}\left[\,{\rm Re}\,\psi\!\left(1+i\frac{\bar\eta_{ab}}{\zeta^{1/2}}\right)-\ln\!\left\{\frac{\bar\eta_{ab}}{\zeta^{1/2}}\right\}\right]\ . \tag{1.5}$$

These expressions are accurate to leading and next-to-leading order in the plasma coupling $g = e^2\kappa/4\pi T$, and are therefore essentially exact in a weakly coupled plasma, *i.e.* a plasma for which $g \ll 1$. For convenience we have set the arbitrary wave number K that appears in the regular and singular terms to the value $K = \kappa_e$.[2] The reduced mass of species a and b is

$$\frac{1}{m_{ab}} = \frac{1}{m_a} + \frac{1}{m_b}\ , \tag{1.6}$$

[1] By convention, $b = 1$ will be the electron component.

[2] Recall that the sum $\mathcal{C}^{\rm CL}_{ab} = \mathcal{C}^{\rm CL}_{ab,{\rm R}} + \mathcal{C}^{\rm CL}_{ab,{\rm S}}$ is independent K. This is because the BPS calculation introduced an arbitrary wave number K which hold no physical significance (it is akin to a renormalization scale in quantum field theory).

while the thermal velocity and the quantum parameter are determined by

$$V_{ab}^2 \;=\; \frac{1}{\beta_a m_a} + \frac{1}{\beta_b m_b} \tag{1.7}$$

$$\bar{\eta}_{ab} \;=\; \frac{e_a e_b}{4\pi\,\hbar V_{ab}}\;. \tag{1.8}$$

The function $F(v)$ takes the form

$$F(v) \;=\; -\int_{-\infty}^{\infty} du\,\frac{\rho_{\rm tot}(u)}{v-u+i\eta} \quad \text{with} \quad \rho_{\rm tot}(u) = \sum_b \rho_b(u) \tag{1.9}$$

$$\rho_b(v) \;=\; \kappa_b^2\sqrt{\frac{\beta_b m_b}{2\pi}}\,v\,\exp\!\left\{-\frac{1}{2}\,\beta_b m_b\,v^2\right\}\;, \tag{1.10}$$

and its relation to the dielectric function is $k^2\,\epsilon(\mathbf{k},\mathbf{k}\cdot\mathbf{v}) = k^2 + F(\hat{\mathbf{k}}\cdot\mathbf{v})$. The first term $\mathcal{C}^{\rm C}_{ab,{\rm R}}$ arises from long-distance collective effects from the dielectric function, and it involves *all* plasma species (even species c different from a and b). This is the term I call non-separable, meaning that it cannot be written as a sum of individual plasma components involving only a single species. The second term $\mathcal{C}^{\rm C}_{ab,{\rm S}}$ arises from short-distance two-body classical scattering, and the third term $\mathcal{C}^{\rm QM}_{ab}$ is the two-body quantum scattering correction to all orders in the quantum parameters $\bar{\eta}_{ab}$. Three body and higher effects are contained in our systematic error term, the next-to-next-to-leading order term proportional to g^3. In a strongly coupled plasma these higher order effects dominate, but in a weakly coupled plasma they are negligible.

A dramatic simplification occurs under the following conditions: (i) the extreme quantum limit is realized, *i.e.* $\bar{\eta}_{ab} \ll 1$, (ii) there is a large mass hierarchy so that $m_e/m_{\rm I} \ll 1$ (in which case the ions have the same temperature $T_{\rm I}$), and (iii) sum over the ions to construct the effective rate coefficient. The rate equation (1.1) becomes

$$\frac{d\mathcal{E}_{e{\rm I}}}{dt} = -\,\mathcal{C}_{e{\rm I}}\left(T_e - T_{\rm I}\right)\,, \tag{1.11}$$

where $d\mathcal{E}_{e{\rm I}}/dt = \sum_i d\mathcal{E}_{ei}/dt$ and $\mathcal{C}_{e{\rm I}} = \sum_i \mathcal{C}_{ei}$. Because of the aforementioned sum-rule and the extreme quantum limit, the result simplifies to

$$\mathcal{C}^{\rm BPS}_{e{\rm I}} = \frac{\kappa_e^2\,\omega_{\rm I}^2}{2\pi}\sqrt{\frac{m_e}{2\pi\,T_e}}\,\ln\Lambda_{\rm BPS}\;, \quad \text{with} \quad \ln\Lambda_{\rm BPS} = \frac{1}{2}\left[\ln\!\left\{\frac{8T_e^2}{\hbar^2\omega_e^2}\right\} - \gamma - 1\right]\;, \tag{1.12}$$

where $\omega_{\rm I}^2 = \sum_i \omega_i^2$. As opposed to the model of Lee-More, there is no ion temperature dependence inside the logarithm. The lack of ion dependence has also been observed by Diamante and Daligault in their classical MD simulations.

II. THE BORN APPROXIMATION

We will code the rate in the Born approximation first:

$$\mathcal{C}_{e\mathrm{I}}^{\mathrm{BPS}} = \underbrace{\frac{\kappa_e^2\,\omega_{\mathrm{I}}^2}{2\pi}\sqrt{\frac{m_e}{2\pi\,T_e}}}_{\mathrm{cn}} \cdot \ln\Lambda_{\mathrm{BPS}}\ , \quad \text{with} \quad \ln\Lambda_{\mathrm{BPS}} = \frac{1}{2}\left[\ln\left\{\frac{8T_e^2}{\hbar^2\omega_e^2}\right\} - \gamma - 1\right]\ . \qquad (2.1)$$

rate.f90:bps_rate_cei_born

```
      SUBROUTINE bps_rate_cei_born(nni, betab, zb, mb, nb, ln_bps_born, cei_born)
      USE bpsvars
      USE mathvars
      USE physvars
      IMPLICIT NONE
      INTEGER,                        INTENT(IN)  :: nni     !  Number of ions
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: betab   !  Temperature array    [1/
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: zb      !  Charge array
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: mb      !  Mass array [keV]
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: nb      !  Number density array [cm

      REAL                       , INTENT(OUT) :: ln_bps_born! BPS Coulomb log
      REAL                       , INTENT(OUT) :: cei_born   ! equilibration rate

      REAL,    PARAMETER :: UPM=0.7745966692E0
      REAL,    PARAMETER :: W13=0.5555555556E0, W2=0.8888888889E0

      REAL    :: mi, ni, zi, omi2, ome2, cn
      REAL    :: me, ne, betae, te2, ke2
      INTEGER :: ib
!
! construct plasma quantities: k_e^2, omega_I^2
!
      ne=nb(1)
      me=mb(1)
      betae=betab(1)
      te2 =1/betae**2
      ke2 =8*PI*AOCM*BEKEV*ne*betae
      ome2=8*PI*AOCM*BEKEV*ne*CC2/me
      omi2=0
      DO ib=2,nni+1
         ni=nb(ib)
         mi=mb(ib)
         zi=zb(ib)
         omi2=omi2 + 8*PI*AOCM*BEKEV*ni*zi*zi*CC2/mi
      ENDDO
!
!                            om_I^2*k_e^2    [ beta_e m_e*c^2 ]^1/2    1
! construct prefactor: cn = ------------  * |---------------- |     * ---
!                               2 pi        [   2 pi         ]        c
!
      cn=omi2*ke2/TWOPI
      cn=cn*SQRT(betae*me/TWOPI)/CC
!
!                                         1   [   [    8 T_e^2    ]             ]
! construct BPS Coulomb log: ln_bps_born = --- [ ln[ ------------- ] - GAMMA -1  ]
!                                         2   [   [  hbar^2 ome^2 ]             ]

      ln_bps_born=0.5*(LOG(8*te2*CC2/(HBARC2*ome2)) -GAMMA - 1)
!
! construct rate: cei_born
!
      cei_born=cn*ln_bps_born

      END SUBROUTINE bps_rate_cei_born
```

III. GENERAL CASE: THE MAIN DRIVER

I will return the general rate coefficients in three forms:

i. `bps_rate_cab_mass`: For a given pair of indices a and b, this subroutines returns the coefficient $\mathcal{C}_{ab}$. The quantum parameter η and the electron mass m_e can be arbitrary. This routine is used to construct the entries in the next two subroutines.

ii. `bps_rate_cab_matrix`: Returns the complete matrix of coefficients $\mathcal{C}_{ab}(E)$.

iii. `bps_rate_cei_mass`: This routine returns the sum over the ions $\mathcal{C}_{e\text{I}} = \sum_i \mathcal{C}_{ei}$. It also returns the coulomb logarithm.

A. The Driver Routine: bps_rate_cab_mass

rate.f90:bps_rate_cab_mass

```
      SUBROUTINE bps_rate_cab_mass(nni, ia, ib, betab, zb, mb, nb, &
        c_ab, c_ab_sing, c_ab_reg, c_ab_qm)
      USE mathvars
      USE physvars
        IMPLICIT NONE
        INTEGER,                                INTENT(IN)  :: nni     !  Number of ions
        INTEGER,                                INTENT(IN)  :: ia      !  Species number
        INTEGER,                                INTENT(IN)  :: ib      !  Species number
        REAL,    DIMENSION(1:nni+1),            INTENT(IN)  :: betab   !  Temperature ar
        REAL,    DIMENSION(1:nni+1),            INTENT(IN)  :: zb      !  Charge array
        REAL,    DIMENSION(1:nni+1),            INTENT(IN)  :: mb      !  Mass array [ke
        REAL,    DIMENSION(1:nni+1),            INTENT(IN)  :: nb      !  Number density
        REAL,                                   INTENT(OUT) :: c_ab
        REAL,                                   INTENT(OUT) :: c_ab_sing
        REAL,                                   INTENT(OUT) :: c_ab_reg
        REAL,                                   INTENT(OUT) :: c_ab_qm
        REAL,    DIMENSION(1:nni+1)  :: kb2, ob2
        REAL    ::  c_s, c_r, c_q
        REAL    :: kia2, kib2, bmia, bmib, nab, omi2, nab_reg
        INTEGER :: ibmax

        ibmax=nni+1
!
! initialize components of rate-coefficients
!
!
! construct plasma quantities: kb^2
!
        ob2=8*PI*AOCM*BEKEV*zb*zb*nb*CC2/mb
        kb2=8*PI*AOCM*BEKEV*zb*zb*nb*betab
        omi2=SUM(ob2(2:ibmax))

        kia2  =kb2(ia)
        bmia  =betab(ia)*mb(ia)
        kib2=kb2(ib)
        bmib=betab(ib)*mb(ib)
        nab=kia2*kib2*CC*SQRT(bmia*bmib)/(bmia + bmib)**1.5   !
        nab=nab/TWOPI**1.5                                    ! normalization
!
! C_{ab}-classical-singular
!
        CALL cab_sing_mass(nni,ia,ib,betab,zb,mb,nb,c_s)
        c_ab_sing=nab*c_s
!
! C_{ab}-classical-regular
```

```
!
        CALL cab_reg_mass(nni,ia,ib,betab,zb,mb,nb,c_r)
        nab_reg=kb2(ia)*omi2/TWOPI
        nab_reg=nab_reg*SQRT(betab(ia)*mb(ia)/TWOPI)/CC
        c_ab_reg=nab_reg*c_r
!
! C_{ab}-quantum
!
        CALL cab_qm_mass(nni,ia,ib,betab,zb,mb,c_q)
        c_ab_qm=nab*c_q
!
! C_{ab}-total
!
        c_ab=c_ab_sing + c_ab_reg + c_ab_qm
      END SUBROUTINE bps_rate_cab_mass
```

rate.f90:bps_rate_cab_matrix

```
      SUBROUTINE bps_rate_cab_matrix(nni, betab, zb, mb, nb, &
            c_ab, c_ab_sing, c_ab_reg, c_ab_qm)
      USE physvars
      USE mathvars
        IMPLICIT NONE                                              ! Plasma:
        INTEGER,                              INTENT(IN)  :: nni   !  number of ions
        REAL,    DIMENSION(1:nni+1),          INTENT(IN)  :: betab !  temp array [1/ke
        REAL,    DIMENSION(1:nni+1),          INTENT(IN)  :: zb    !  charge array
        REAL,    DIMENSION(1:nni+1),          INTENT(IN)  :: mb    !  mass array [keV]
        REAL,    DIMENSION(1:nni+1),          INTENT(IN)  :: nb    !  density [1/cc]
                                                                   !
                                                                   ! A-coeffs [MeV/mic
        REAL,    DIMENSION(1:nni+1,1:nni+1),INTENT(OUT) :: c_ab
        REAL,    DIMENSION(1:nni+1,1:nni+1),INTENT(OUT) :: c_ab_sing
        REAL,    DIMENSION(1:nni+1,1:nni+1),INTENT(OUT) :: c_ab_reg
        REAL,    DIMENSION(1:nni+1,1:nni+1),INTENT(OUT) :: c_ab_qm
        REAL    :: cab, cab_sing, cab_reg, cab_qm
        INTEGER :: ia, ib
        DO ia=1,nni+1
          DO ib=1,nni+1
          CALL bps_rate_cab_mass(nni, ia, ib, betab, zb, mb, nb, &
            cab, cab_sing, cab_reg,cab_qm)
            c_ab(ia,ib)=cab
            c_ab_sing(ia,ib)=cab_sing
            c_ab_reg(ia,ib) =cab_reg
            c_ab_qm(ia,ib)  =cab_qm
          ENDDO
        ENDDO
      END SUBROUTINE bps_rate_cab_matrix
```

rate.f90:bps_rate_cei_mass

```
      SUBROUTINE bps_rate_cei_mass(nni, betab, zb, mb, nb, ln_bps,      &
      delta, cei_tot, cei_i, cei_e, ceic_tot, ceic_i, ceic_e, ceiq_tot, &
      ceiq_i, ceiq_e, ceic_s_i, ceic_s_e, ceic_r_i , ceic_r_e, ceib)
      USE physvars
      USE mathvars
        IMPLICIT NONE
        INTEGER,                        INTENT(IN)  :: nni     !  Number of ions
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: betab   !  Temperature array    [
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: zb      !  Charge array
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: mb      !  Mass array [keV]
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: nb      !  Number density array [
        REAL,                        INTENT(OUT) :: ln_bps  !  Coulomb logarithm
        REAL,                        INTENT(OUT) :: delta   !  C_reg =-1/2+delta
```

```
        REAL,                          INTENT(OUT) :: cei_tot ! [cm^-3 s^-1]
        REAL,                          INTENT(OUT) :: cei_i   !
        REAL,                          INTENT(OUT) :: cei_e   !
        REAL,                          INTENT(OUT) :: ceic_tot!
        REAL,                          INTENT(OUT) :: ceic_i  !
        REAL,                          INTENT(OUT) :: ceic_e  !
        REAL,                          INTENT(OUT) :: ceiq_tot!
        REAL,                          INTENT(OUT) :: ceiq_i  !
        REAL,                          INTENT(OUT) :: ceiq_e  !
        REAL,                          INTENT(OUT) :: ceic_s_i!
        REAL,                          INTENT(OUT) :: ceic_s_e!
        REAL,                          INTENT(OUT) :: ceic_r_i!
        REAL,                          INTENT(OUT) :: ceic_r_e!
        REAL,    DIMENSION(1:nni+1), INTENT(OUT) :: ceib    !
        REAL,    DIMENSION(1:nni+1)  :: ob2
        REAL     :: omi2, cn, kb2e
        REAL     :: cab, c_ab_sing, c_ab_reg, c_ab_qm
        INTEGER  :: ia, ib, nnb
!
! initialize components of A-coefficients
!
        delta=0
        cei_tot =0  ! electron + ion
        cei_i   =0  ! ion contribution
        cei_e   =0  ! electron contribution
        ceic_tot=0  ! classical total
        ceic_e  =0  ! classical electron
        ceic_i  =0  ! classical ion
        ceiq_tot=0  ! quantum total
        ceiq_e  =0  ! quantum electron
        ceiq_i  =0  ! quantum ion
        ceic_s_i=0
        ceic_s_e=0
        ceic_r_i=0
        ceic_r_e=0
        ceib=0

        kb2e=8*PI*AOCM*BEKEV*zb(1)*zb(1)*nb(1)*betab(1)
        ob2=8*PI*AOCM*BEKEV*zb*zb*nb*CC2/mb
        omi2=SUM(ob2(2:nni+1))
        cn=kb2e*omi2/TWOPI
        cn=cn*SQRT(betab(1)*mb(1)/TWOPI)/CC

        ia=1
        NNB = nni+1    ! number of ions + electrons
        DO ib=1,nni+1  ! loop over electrons to calculate cei_e
           CALL bps_rate_cab_mass(nni, ia, ib, betab, zb, mb, nb, &
           cab, c_ab_sing, c_ab_reg, c_ab_qm)
           IF (ib .NE. 1) delta=delta+c_ab_reg/cn
           ceib(ib)=c_ab_sing + c_ab_reg + c_ab_qm
           CALL x_collect(ib, NNB, c_ab_sing, c_ab_reg, c_ab_qm,      &
           cei_tot, cei_i, cei_e, ceic_tot, ceic_i, ceic_e, ceiq_tot, &
           ceiq_i, ceiq_e, ceic_s_i, ceic_s_e, ceic_r_i, ceic_r_e)
        ENDDO
        delta=0.5+delta
        ln_bps=cei_i/cn
      END SUBROUTINE bps_rate_cei_mass
```

The quantity Δ in the last subroutine is now described. To obtain the leading order term in the electron mass, we can employ the sum rule

$$\frac{i}{2\pi}\int_{-\infty}^{\infty} dv\, v\left[\kappa_e^2 + F_{\rm I}(v)\right]\ln\left\{1+\frac{F_{\rm I}(v)}{\kappa_e^2}\right\} = -\frac{1}{2}\sum\nolimits_i \omega_i^2\ , \tag{3.1}$$

which gives (the superscript denotes leading order in m_e)

$$\mathcal{C}^0_{e\mathrm{I,R}} \;=\; -\frac{\kappa_e^2}{2\pi}\left(\frac{\beta_e m_e}{2\pi}\right)^{1/2}\frac{\omega_{\mathrm{I}}^2}{2} \quad \text{with} \;\; \omega_{\mathrm{I}}^2 = \sum\nolimits_i \omega_i^2 \;. \tag{3.2}$$

We will express the rate coefficient (3.11) in terms of a correction Δ that vanishes in the $m_e \to 0$ limit:

$$\mathcal{C}_{e\mathrm{I,R}} \;=\; \frac{\kappa_e^2}{2\pi}\left(\frac{\beta_e m_e}{2\pi}\right)^{1/2}\omega_{\mathrm{I}}^2\Big[-\frac{1}{2}+\Delta\Big]\;, \tag{3.3}$$

with

$$\Delta \;\equiv\; \frac{1}{2}+\frac{1}{\omega_{\mathrm{I}}^2}\frac{i}{2\pi}\int_{-\infty}^{\infty} dv\, v\, e^{-\frac{1}{2}\beta_e m_e v^2}\frac{\rho_{\mathrm{I}}(v)}{\rho_{\mathrm{tot}}(v)}\,F(v)\ln\!\left\{\frac{F(v)}{\kappa_e^2}\right\} \tag{3.4}$$

$$=\; \frac{1}{\omega_{\mathrm{I}}^2}\frac{i}{2\pi}\int_{-\infty}^{\infty} dv\, v\left[e^{-\frac{1}{2}\beta_e m_e v^2}\frac{\rho_{\mathrm{I}}(v)}{\rho_{\mathrm{tot}}(v)}\,F(v)\ln\!\left\{\frac{F(v)}{\kappa_e^2}\right\} - \Big[\kappa_e^2+F_{\mathrm{I}}(v)\Big]\ln\!\left\{1+\frac{F_{\mathrm{I}}(v)}{\kappa_e^2}\right\}\right]. \tag{3.5}$$

The form (3.5) can serve as a starting point for an analytic calculation (see notes: this doesn't seem to lead to particularly simple results). For numerical work it is more useful to compute the integral in (3.4) directly (using some quadrature method), and if desired, we may construct Δ from (3.3).

B. The Singular Contribution: cab_sing_mass

From Eq. (1.4) of Section I the short distance contribution to the total electron-ion rate from the singular piece

$$\mathcal{C}^{\mathrm{C}}_{ab,\mathrm{S}} \;=\; -\kappa_a^2\,\kappa_b^2\,\frac{(\beta_a m_a \beta_b m_b)^{1/2}}{(\beta_a m_a+\beta_b m_b)^{3/2}}\left(\frac{1}{2\pi}\right)^{3/2}\left[\ln\!\left\{\frac{e_a\,e_b}{4\pi}\,\frac{\kappa_e}{4\,m_{ab}\,V_{ab}^2}\right\}+2\gamma\right], \tag{3.6}$$

where

$$\frac{1}{m_{ab}} \;=\; \frac{1}{m_a}+\frac{1}{m_b} \tag{3.7}$$

$$V_{ab}^2 \;=\; \frac{1}{\beta_a m_a}+\frac{1}{\beta_b m_b}\;. \tag{3.8}$$

It will prove more convenient to use the form

$$\mathcal{C}^{\mathrm{C}}_{e\mathrm{I,S}} \;=\; \underbrace{\kappa_a^2\,\kappa_b^2\,\frac{(\beta_a m_a \beta_b m_b)^{1/2}}{(\beta_a m_a+\beta_b m_b)^{3/2}}\left(\frac{1}{2\pi}\right)^{3/2}}_{\mathrm{nab}}\cdot \mathsf{C}_{\mathsf{ab},\mathrm{S}} \tag{3.9}$$

$$\mathsf{C}_{\mathsf{ab},\mathrm{S}} \;=\; -\ln\!\left\{\frac{g_e\,Z_a Z_b}{4\beta_e\,m_{ab}V_{ab}^2}\right\}-2\gamma\;. \tag{3.10}$$

The subroutine below gives C_{ab} and the coefficient resides in the driver or calling routine.

rate.f90:cab_sing_mass

```
      SUBROUTINE cab_sing_mass(nni, ia, ib, betab, zb, mb, nb, cab_sing)
      USE mathvars
      USE physvars
        IMPLICIT NONE
        INTEGER,                      INTENT(IN)  :: nni  !Number of ion species
        INTEGER,                      INTENT(IN)  :: ia
        INTEGER,                      INTENT(IN)  :: ib
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: zb   !Charge array
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: betab!Temperature array [1/keV]
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: mb   !Mass array [keV]
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: nb   !density array [1/cm^3]
        REAL,                        INTENT(OUT) :: cab_sing

        REAL    :: betae, ne, ge, mabc2, zia, zib, vab2
!
! note: om_b=(1.32155E+3)*SQRT(zb*zb*nb*AMUKEV/mb) ! Plasma frequency [1/s]
!       ge=(6.1260E-15)*SQRT(ne)/te**1.5

        betae=betab(1)
        ne=nb(1)
        ge=GECOEFF*SQRT(ne)*betae**1.5
        zia=zb(ia)
        zib=zb(ib)
        mabc2=mb(ia)*mb(ib)/(mb(ia) + mb(ib)) ! [keV]
        vab2=1./(betab(ia)*mb(ia)) + 1./(betab(ib)*mb(ib)) ! [dimensionless]
        cab_sing=-LOG(0.25*ge*ABS(zia*zib)/(mabc2*betae*vab2)) - 2*GAMMA
      END SUBROUTINE cab_sing_mass
```

C. The Regular Contribution: cab_reg_mass

From Section I, the long distance regular piece of the electron-ion equilibration rate is given by

$$\mathcal{C}^{\rm C}_{ab,{\rm R}} \;=\; \frac{\kappa_a^2}{2\pi}\left(\frac{\beta_a m_a}{2\pi}\right)^{1/2}\frac{i}{2\pi}\int_{-\infty}^{\infty} dv\, v\, e^{-\frac{1}{2}\beta_a m_a v^2}\,\frac{\rho_b(v)}{\rho_{\rm tot}(v)}\,F(v)\ln\left\{\frac{F(v)}{\kappa_e^2}\right\} \tag{3.11}$$

$$\;=\; -\frac{\kappa_a^2}{2\pi}\left(\frac{\beta_a m_a}{2\pi}\right)^{1/2}\frac{1}{2\pi}\int_{0}^{\infty} dv\, v\, e^{-\frac{1}{2}\beta_a m_a v^2}\,\frac{\rho_b(v)}{\rho_{\rm tot}(v)}\,H(v)\;, \tag{3.12}$$

where we have defined

$$H(v) \;\equiv\; -i\left[F(v)\ln\left\{\frac{F(v)}{\kappa_e^2}\right\} - F^*(v)\ln\left\{\frac{F^*(v)}{\kappa_e^2}\right\}\right] = 2\left[F_{\rm Re}\arg\{F\} + F_{\rm Im}\ln\left\{\frac{|F|}{\kappa_e^2}\right\}\right]\;. \tag{3.13}$$

Equation (3.12) follows from expression (3.11) upon using the relation $F(-v) = F^*(v)$ in the negative v-regime. To calculate Δ it is convenient to express the regular piece in the form

$$\mathcal{C}^{\rm C}_{ab,{\rm R}} \;=\; \underbrace{\frac{\kappa_a^2}{2\pi}\left(\frac{\beta_a m_a}{2\pi}\right)^{1/2}\omega_{\rm I}^2}_{\rm nab\_reg}\cdot \mathsf{C}_{\mathsf{ab},{\rm R}} \tag{3.14}$$

$$\mathsf{C}_{\mathsf{ab},{\rm R}} \;=\; -\frac{1}{2\pi\,\omega_{\rm I}^2}\int_{0}^{\infty} dv\, v\, e^{-\frac{1}{2}\beta_a m_a v^2}\,\frac{\rho_b(v)}{\rho_{\rm tot}(v)}\,H(v)\;. \tag{3.15}$$

We will rescale the velocity integration to form the dimensionless variable

$$x = \frac{v\,\mu}{\sqrt{2}} \qquad \mu = \sum\nolimits_i \beta_i m_i \tag{3.16}$$

and defining $\mathbb{H}(x) = H(v)/\kappa_e^2$ gives

$$\mathsf{C}_{\mathsf{ab,R}} \;=\; -\frac{\kappa_e^2}{\pi\,\mu^2\,\omega_{\rm I}^2}\int_0^\infty dx\,x\,e^{-\beta_a m_a x^2/\mu^2}\,\frac{\kappa_b^2(\beta_b m_b)^{1/2}e^{-\beta_b m_b x^2/\mu^2}}{\sum_c \kappa_c^2(\beta_c m_c)^{1/2}e^{-\beta_c m_c x^2/\mu^2}}\,\mathbb{H}(x) \tag{3.17}$$

$$\;=\; -\frac{\kappa_e^2}{\pi\,\mu^2\,\omega_{\rm I}^2}\int_0^\infty dx\,\left[\sum_c \frac{\kappa_c^2\,(\beta_c m_c)^{1/2}}{\kappa_b^2\,(\beta_b m_b)^{1/2}}\,e^{(\beta_a m_a+\beta_b m_b-\beta_c m_c)x^2/\mu^2}\right]^{-1}\mathbb{H}(x)\,x\;. \tag{3.18}$$

The quantity passed to the subroutine `frfi` is

$$a_b = \left(\frac{\beta_b m_b}{2}\right)^{1/2} v = (\beta_b m_b)^{1/2}\,\frac{x}{\mu} \tag{3.19}$$

rate.f90:

```
      FUNCTION dcab_reg(x, nni, ia, ib, betab, zb, mb, nb)
      USE physvars
      USE mathvars
      IMPLICIT NONE
      REAL,                          INTENT(IN)  :: x
      INTEGER,                       INTENT(IN)  :: nni    !  Number of ion species
      INTEGER,                       INTENT(IN)  :: ia     !  Species type
      INTEGER,                       INTENT(IN)  :: ib     !  Species type
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: zb     !  Charge array
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: betab  !  Temperature array [1/keV]
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: mb     !  Mass array [keV]
      REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: nb     !  density array [1/cm^3]
      REAL                                     :: dcab_reg
      REAL,    DIMENSION(1:nni+1) :: ab, kbar2b
      REAL                        :: fr, fi, fabs, farg
      REAL                        :: mu, mu2, rx, hx, ne, betae, cn, kaic, kaia, abca
      INTEGER                     :: ic
!
! construct parameters
!
      mu=SUM(SQRT(betab(2:nni+1)*mb(2:nni+1)))  ! inverse thermal velocity
      mu2=mu*mu                                 ! mu^2
      ab=SQRT(betab*mb)/mu                      !

      ne=nb(1)                                  !
      betae=betab(1)                            !
      kbar2b=zb*zb*betab*nb/(betae*ne)          ! kbar=k_b/k_e
!
! construct H(x)*x from  F_re, F_im, |F|, arg(F)
!
      CALL frfi(x,nni,kbar2b,ab,fr,fi,fabs,farg)! F(x)
      hx=2*(fi*LOG(fabs) + fr*farg)*x           ! H(x)*x
!
! construct spectral weight ratio R_ab(x)

      rxic=0
      DO ic=1,nni+1
         kaic=kbar2b(ic)*ab(ic)
         IF (ic == ib) kaia=kaic*EXP(-ab(ib)*ab(ib)*x*x)
         abca2 =ab(ic)**2 - ab(ia)**2
```

```
         rxic = rxic + kaic*EXP(-abca2*x*x)
      ENDDO
      rx=kaia/rxic
!
! construct un-normalized rate integrand
!
      dcab_reg=rx*hx
!
! construct normalization coefficient
!
      cn=0
      DO ic=2,nni+1                                                    !
         cn = cn + zb(ic)*zb(ic)*nb(ic)/(mb(ic)*ne*betae)              ! omega_I^2/kappa_e^2
      ENDDO                                                            !
      cn=-1./(PI*mu2*cn)                                               ! ke^2/(PI*mu^2 omI^2)
      dcab_reg=dcab_reg*cn

      END FUNCTION dcab_reg
      SUBROUTINE cab_reg_mass(nni, ia, ib, betab, zb, mb, nb, c_r)
        IMPLICIT NONE
        INTEGER,                      INTENT(IN)  :: nni    !  Number of ion species
        INTEGER,                      INTENT(IN)  :: ia     !
        INTEGER,                      INTENT(IN)  :: ib     !
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: zb     !  Charge array
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: betab  !  Temperature array [1/ke
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: mb     !  Mass array [keV]
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: nb     !  density array [1/cm^3]
        REAL,                        INTENT(OUT) :: c_r

        REAL,    PARAMETER :: UPM=0.7745966692E0
        REAL,    PARAMETER :: W13=0.5555555556E0, W2=0.8888888889E0

        REAL    :: y, dcab_reg
        REAL    :: xmin, xmax, x, dx, xm, xc
        INTEGER :: nmax, ic
!
! integration cutoff determined by thermal velocity of ions
!
        xc=2. ! ti=1, te=0.01, 0.1, 1, 10, 100
        xmin=0.
        xmax=5*xc ! automate this choice later.
        nmax=1000
        dx=(xmax-xmin)/nmax
        x=xmin-dx
        c_r=0
        DO ic=1,nmax,2
!
           x=x+2.E0*dx
           y=dcab_reg(x,nni,ia,ib,betab,zb,mb,nb)
           c_r=c_r + W2*y
!
           xm=x-dx*UPM
           y=dcab_reg(xm,nni,ia,ib,betab,zb,mb,nb)
           c_r=c_r + W13*y
!
           xm=x+dx*UPM
           y=dcab_reg(xm,nni,ia,ib,betab,zb,mb,nb)
           c_r=c_r + W13*y
        ENDDO
        c_r=c_r*dx
      END SUBROUTINE cab_reg_mass
```

***===

D. The Quantum Correction: cab_qm_mass

From Eq. (1.5) of Section I, the quantum correction is

$$\mathcal{C}_{ab}^{\rm QM} = \underbrace{\kappa_a^2\,\kappa_b^2\,\frac{(\beta_a m_a\,\beta_b m_b)^{1/2}}{(\beta_a m_a+\beta_b m_b)^{3/2}}\left(\frac{1}{2\pi}\right)^{3/2}}_{\rm nab} \times \underbrace{-\frac{1}{2}\int_0^\infty d\zeta\, e^{-\zeta/2}\left[\,{\rm Re}\,\psi\left(1+i\frac{\bar\eta_{ab}}{\zeta^{1/2}}\right)-\ln\left\{\frac{\bar\eta_{ab}}{\zeta^{1/2}}\right\}\right]}_{\rm c\_q}\ . \tag{3.20}$$

rate.f90:

```
      FUNCTION dcei_qm(x, eta)
      IMPLICIT NONE
      REAL  :: x       ! xi
      REAL  :: eta     ! quantum parameter
      REAL  :: dcei_qm ! quantum integrand
      REAL  :: xh, repsilog
      xh=eta/SQRT(x)
      dcei_qm=EXP(-x/2)*repsilog(xh)
      END FUNCTION dcei_qm


      SUBROUTINE cab_qm_mass(nni, ia, ib, betab, zb, mb, qm)
      USE physvars
      USE mathvars
        IMPLICIT NONE
        INTEGER,                      INTENT(IN)  :: nni    !  Number of ion species
        INTEGER,                      INTENT(IN)  :: ia     !
        INTEGER,                      INTENT(IN)  :: ib     !
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: zb     !  Charge array
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: betab  !  Temperature array [1/ke
        REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: mb     !  Mass array [keV]
        REAL,                        INTENT(OUT) :: qm

        REAL,    PARAMETER :: UPM=0.7745966692E0
        REAL,    PARAMETER :: W13=0.5555555556E0, W2=0.8888888889E0
        INTEGER, PARAMETER :: NMAX=1000
        REAL    :: xmin, xmax, dx, x, xm, y, dcei_qm, eta_ab, vab
        INTEGER :: ix

        vab=SQRT(1./(betab(ia)*mb(ia)) + 1./(betab(ib)*mb(ib)))
        eta_ab=ABS(zb(ia)*zb(ib))*2*BEKEV*A0CM/HBARC/vab

        xmin=0.
        xmax=15
        dx=(xmax-xmin)/NMAX
        x=xmin-dx
        qm=0
        DO ix=1,NMAX,2
!
           x=x+2*dx
           y=dcei_qm(x, eta_ab)
           qm=qm + W2*y
!
           xm=x-dx*UPM
           y=dcei_qm(xm,eta_ab)
           qm=qm + W13*y
!
           xm=x+dx*UPM
           y=dcei_qm(xm,eta_ab)
           qm=qm + W13*y
        ENDDO
        qm=-0.5*qm*dx
      END SUBROUTINE cab_qm_mass
```

Appendix A: Calculating the Real and Imaginary Parts of F

We can write the dielectric function (1.9) as a sum over plasma components,

$$F(v) = \sum\nolimits_b F_b(v) \ , \tag{A1}$$

where we express the contribution from plasma species b as

$$F_b(v) \ = \ -\int_{-\infty}^{\infty} du \, \frac{\rho_b(v)}{v - u + i\eta} \tag{A2}$$

$$\rho_b(v) \ = \ \kappa_b^2 \sqrt{\frac{\beta_b m_b}{2\pi}} \, v \, \exp\!\left\{ -\frac{1}{2} \, \beta_b m_b \, v^2 \right\} \ . \tag{A3}$$

We will often decompose F into its contribution from electrons and ions and write

$$F(v) = F_e(v) + F_{\rm I}(v) \ . \tag{A4}$$

Note the reflection property

$$F_b(-v) \ = \ F_b^*(v) \ , \tag{A5}$$

which means that the real part of $F_b(v)$ is even in v and the imaginary part is odd. For numerical work it is best to use the explicit real and imaginary parts of F, which can be written

$$F_{\rm Re}(v) \ = \ \sum_b \kappa_b^2 \left[1 - 2\sqrt{\frac{\beta_b m_b}{2}} \, v \, \mathrm{daw}\left\{ \sqrt{\frac{\beta_b m_b}{2}} \, v \right\} \right] \tag{A6}$$

$$F_{\rm Im}(v) \ = \ \sqrt{\pi} \sum_b \kappa_b^2 \sqrt{\frac{\beta_b m_b}{2}} \ \ v \, \exp\left\{ -\frac{\beta_b m_b}{2} \, v^2 \right\} = \pi \rho_{\rm tot}(v) \ , \tag{A7}$$

where the Dawson integral is defined by

$$\mathrm{daw}(x) = \int_0^x dy \, e^{y^2 - x^2} = \frac{\sqrt{\pi}}{2} \, e^{-x^2} \mathrm{erfi}(x) \ . \tag{A8}$$

The limits of small and large arguments of the Dawson function are

$$\mathrm{daw}(x) \ = \ x + \frac{2x^3}{3} + \frac{4x^5}{15} + \mathcal{O}(x^7) \tag{A9}$$

$$\mathrm{daw}(x) \ = \ \frac{1}{2x} + \frac{1}{4x^3} + \frac{3}{8x^5} + \mathcal{O}(x^{-7}) \ . \tag{A10}$$

The functions $F_b(v)$ have units of wave-number-squared $[1/L^2]$ and their argument v has units of velocity. We can express the functions $F_b(v)$ in terms of a single dimensionless function $\mathbb{F}(x)$ as follows:

$$F_b(v) \ = \ \kappa_b^2 \, \mathbb{F}\left(\sqrt{\frac{\beta_b m_b}{2}} \, v \ \right) \tag{A11}$$

with

$$\mathbb{F}(x) \;=\; \int_{-\infty}^{\infty} dy\, \frac{\bar\rho(y)}{y-x-i\eta} \tag{A12}$$

$$\bar\rho(y) \;=\; \frac{1}{\sqrt{\pi}}\, y\, e^{-y^2} \;. \tag{A13}$$

Relation (A11) holds because $\rho_b(u) = \kappa_b^2\, \bar\rho(y)$ for $u = (2/\beta_b m_b)^{1/2}\, y$. The reflection property (A5) becomes

$$\mathbb{F}(-x) = \mathbb{F}^*(x) \;, \tag{A14}$$

which means that the real part is even in x and the imaginary part is odd,

$$\mathbb{F}_{\rm Re}(-x) \;=\; \;\; \mathbb{F}_{\rm Re}(x) \tag{A15}$$

$$\mathbb{F}_{\rm Im}(-x) \;=\; -\mathbb{F}_{\rm Im}(x) \;. \tag{A16}$$

As with expressions (A6) and (A7), the real and imaginary parts can be written

$$\mathbb{F}_{\rm Re}(x) \;=\; 1 - 2x\, {\rm daw}\,(x) \tag{A17}$$

$$\mathbb{F}_{\rm Im}(x) \;=\; \pi\, \bar\rho(x) = \sqrt{\pi}\, x\, e^{-x^2} \;. \tag{A18}$$

Let us now establish the forms (A6) and (A7) for the real and imaginary parts of $F(v)$. Staring with

$$\frac{1}{y-x-i\eta} = \mathsf{P}\frac{1}{y-x} + i\pi\, \delta(y-x) \;, \tag{A19}$$

the imaginary part becomes

$$\mathbb{F}_{\rm Im}(x) \;=\; \int_{-\infty}^{\infty} dy\, {\rm Im}\, \frac{\bar\rho(y)}{y-x-i\eta} = \int_{-\infty}^{\infty} dy\, \bar\rho(y)\, \pi\delta(y-x) = \pi\bar\rho(x) \;. \tag{A20}$$

The real part of the function must be evaluated by a principal part integral,

$$\mathbb{F}_{\rm Re}(x) \;=\; \mathsf{P} \int_{-\infty}^{\infty} dy\, \frac{\bar\rho(y)}{y-x} \;. \tag{A21}$$

Let us add and subtract unity in the form

$$\int_{-\infty}^{\infty} \frac{dy}{y}\, \bar\rho(y) = 1 \;, \tag{A22}$$

so that

$$\mathbb{F}_{\rm Re}(x) \;=\; 1 + \mathsf{P} \int_{-\infty}^{\infty} dy\, \left[\frac{\bar\rho(y)}{y-x} - \frac{\bar\rho(y)}{y} \right] \tag{A23}$$

$$\;=\; 1 + \mathsf{P} \int_{-\infty}^{\infty} dy\, \frac{x}{y(y-x)}\, \bar\rho(y) \tag{A24}$$

$$\;=\; 1 + \frac{x}{\sqrt{\pi}}\, \mathsf{P} \int_{-\infty}^{\infty} dy\, \frac{e^{-y^2}}{y-x} \;. \tag{A25}$$

Making the change of variables $y' = y - x$ (and dropping the prime) we can write

$$\mathbb{F}_{\rm Re}(x) \;=\; 1 + \frac{x}{\sqrt{\pi}}\,\mathsf{P}\int_{-\infty}^{\infty}\frac{dy}{y}\,e^{-(y+x)^2} \tag{A26}$$

$$\;=\; 1 + \frac{x\,e^{-x^2}}{\sqrt{\pi}}\,\mathsf{P}\int_{-\infty}^{\infty}\frac{dy}{y}\,e^{-y^2-2xy} \tag{A27}$$

$$\;=\; 1 + \frac{x\,e^{-x^2}}{\sqrt{\pi}}\lim_{\epsilon\to 0^+}\left[\int_{-\infty}^{-\epsilon}\frac{dy}{y}\,e^{-y^2-2xy} + \int_{\epsilon}^{\infty}\frac{dy}{y}\,e^{-y^2-2xy}\right]\,. \tag{A28}$$

In the last expression we have used the definition of the principal part integration. Making a change of variables $y' = -y$ in the first integral in square brackets gives (and again dropping the prime)

$$\int_{-\infty}^{-\epsilon}\frac{dy}{y}\,e^{-y^2-2xy} = -\int_{\epsilon}^{\infty}\frac{dy}{y}\,e^{-y^2+2xy}\;, \tag{A29}$$

and this allows us to write

$$\mathbb{F}_{\rm Re}(x) \;=\; 1 - \frac{x\,e^{-x^2}}{\sqrt{\pi}}\lim_{\epsilon\to 0^+}\int_{\epsilon}^{\infty}\frac{dy}{y}\,e^{-y^2}\left[e^{2xy} - e^{-2xy}\right]\,. \tag{A30}$$

The term in square braces is just $2\sinh(2xy)$, which renders the factor $1/y$ harmless when the limit $\epsilon \to 0^+$ is taken,

$$\mathbb{F}_{\rm Re}(x) \;=\; 1 - \frac{2x\,e^{-x^2}}{\sqrt{\pi}}\int_0^{\infty} dy\,e^{-y^2}\,\frac{\sinh 2xy}{y} = 1 - 2x\,\mathrm{daw}(x)\;. \tag{A31}$$

The latter form hold because this is just another integral representation of the Dawson function,

$$\mathrm{daw}(x) \;=\; \frac{e^{-x^2}}{\sqrt{\pi}}\int_0^{\infty} dy\,e^{-y^2}\,\frac{\sinh 2xy}{y}\;. \tag{A32}$$

Compare this with

$$\mathrm{daw}(x) \;=\; e^{-x^2}\int_0^{x} dy\,e^{y^2}\;. \tag{A33}$$

Appendix B: Proving the Sum Rule

Let's review the calculation of Eq. (3.2). In taking the zero mass limit $m_e \to 0$ of Eq. (3.11), we can replace

$$e^{-\frac{1}{2}\beta_e m_e v^2} \to 1 \tag{B1}$$

This is because the term $\rho_{\rm I}(v)$ and the logarithmic term provide the necessary convergence.[3] In this limit we can also can replace the electron spectral weight by

$$\rho_e(v) \to \kappa_e^2 \left(\frac{\beta_e m_e}{2\pi}\right)^{1/2} v \ , \tag{B2}$$

and we can therefore substitute

$$\frac{\rho_{\rm I}(v)}{\rho_{\rm tot}(v)} = \frac{\rho_{\rm I}(v)}{\rho_{\rm I}(v)+\rho_e(v)} \to \frac{\rho_{\rm I}(v)}{\rho_{\rm I}(v)+\kappa_e^2\,(\beta_e m_e/2\pi)^{1/2}\,v} \ . \tag{B3}$$

In fact, since the convergence is supplied by the logarithmic term, we can take the $m_e \to 0$ limit in the second term of the integrand,

$$\frac{\rho_{\rm I}(v)}{\rho_{\rm I}(v)+\kappa_e^2\,(\beta_e m_e/2\pi)^{1/2}\,v} \to 1 \ , \tag{B4}$$

which allows us to write

$$\bar{\mathcal{C}}^0_{e{\rm I},{\rm R}} = \int_{-\infty}^{\infty} dv\, v\, \frac{\rho_{\rm I}(v)}{\rho_{\rm I}(v)+\kappa_e^2\,(\beta_e m_e/2\pi)^{1/2}\,v} \frac{i}{2\pi}\, F(v) \ln\!\left\{\frac{F(v)}{\kappa_e^2}\right\} \tag{B5}$$

$$= \int_{-\infty}^{\infty} dv\, v\, \frac{i}{2\pi}\, F(v) \ln\!\left\{\frac{F(v)}{\kappa_e^2}\right\} \ . \tag{B6}$$

We can expand the dielectric function as $F(v) = F_e(v) + F_{\rm I}(v)$, where the ion contribution is

$$F_{\rm I}(v) = \int_{-\infty}^{\infty} du\, \frac{\rho_{\rm I}(u)}{u - v - i\eta} \tag{B7}$$

and the electron contribution can be written

$$F_e(v) = \int_{-\infty}^{\infty} du\, \frac{\rho_e(u)}{u - v - i\eta} = \kappa_e^2 \left(\frac{\beta_e m_e}{2\pi}\right)^{1/2} \int_{-\infty}^{\infty} du\, \frac{u\, \exp\{-\frac{1}{2}\,\beta_e m_e u^2\}}{u - v - i\eta} \ . \tag{B8}$$

The support in u comes from $\beta_e m_e \bar{u}_e^2 \sim 1$, while we take the argument v to lie in the region about $\bar{v}_{\rm I}$ determined by $\beta_{\rm I} m_{\rm I} \bar{v}_{\rm I}^2 \sim 1$. In other words, the typical velocity u is the electron thermal velocity, which is much faster than the typical ion velocity velocity v, that is to say $0 < v \ll u$. This means to leading order we can replace $u/(u - v - i\eta) \to 1$; therefore,

$$F_e(v) = \kappa_e^2 \left(\frac{\beta_e m_e}{2\pi}\right)^{1/2} \int_{-\infty}^{\infty} du\, \exp\{-\frac{1}{2}\,\beta_e m_e u^2\} = \kappa_e^2 \ . \tag{B9}$$

[3] The support of the integrand lies near $\bar{v}_{\rm I}$ determined by $\beta_{\rm I} m_{\rm I} \bar{v}_{\rm I}^2 \sim 1$, and for velocities in this vicinity, we see that $0 < \frac{1}{2}\, m_e \beta_e v^2 \ll 1$ when $\beta_e m_e \ll \beta_{\rm I} m_{\rm I}$. We can therefore replace the exponential by 1 for $v \sim \bar{v}_{\rm I}$.

To leading order we can express the total dielectric function as

$$F(v) = \kappa_e^2 + F_{\rm I}(v) \ , \tag{B10}$$

and therefore

$$\bar{\mathcal{C}}^0_{e{\rm I},{\rm R}} \ = \ \int_{-\infty}^{\infty} dv\, v\, \frac{i}{2\pi} \left[\kappa_e^2 + F_{\rm I}(v)\right] \ln\!\left\{1 + \frac{F_{\rm I}(v)}{\kappa_e^2}\right\} \ . \tag{B11}$$

$$= \ \lim_{V\to\infty} \int_{-V}^{V} dv\, v\, \frac{i}{2\pi} \left[\kappa_e^2 + F_{\rm I}(v)\right] \ln\!\left\{1 + \frac{F_{\rm I}(v)}{\kappa_e^2}\right\} \ . \tag{B12}$$

Note that the integrand is analytic in the upper half plane, which means that its integral along any closed path in the upper half plane will vanish,

$$\oint dz\, z\, \frac{i}{2\pi} \left[\kappa_e^2 + F_{\rm I}(z)\right] \ln\!\left\{1 + \frac{F_{\rm I}(z)}{\kappa_e^2}\right\} = 0 \ . \tag{B13}$$

Let C_V denote the counter-clockwise circle at the origin with radius V, and we shall consider the closed contour defined by traversing the real axis along $[-V, V]$ and then along C_V in the the upper half plane. Expression (B13) can then be written

$$\int_{-V}^{V} dv\, v\, \frac{i}{2\pi} \left[\kappa_e^2 + F_{\rm I}(v)\right] \ln\!\left\{1 + \frac{F_{\rm I}(v)}{\kappa_e^2}\right\} + \int_{C_V} dz\, z\, \frac{i}{2\pi} \left[\kappa_e^2 + F_{\rm I}(z)\right] \ln\!\left\{1 + \frac{F_{\rm I}(z)}{\kappa_e^2}\right\} = 0 \ . \tag{B14}$$

Taking the $V \to \infty$ limit gives

$$\bar{\mathcal{C}}^0_{e{\rm I},{\rm R}} \ = \ - \lim_{V\to\infty} \int_{C_V} dz\, z\, \frac{i}{2\pi} \left[\kappa_e^2 + F_{\rm I}(z)\right] \ln\!\left\{1 + \frac{F_{\rm I}(z)}{\kappa_e^2}\right\} \ . \tag{B15}$$

Complex numbers on C_V take the form $z = Ve^{i\theta}$. This means that along C_V we find $dz = iz\, d\theta$ (or $dz/z = id\theta$) and $|z| = V$. To complete the calculation we need $F_{\rm I}(z)$ for large values of $|z|$, which takes the form

$$F_{\rm I}(z) \ = \ -\frac{1}{z^2} \sum\nolimits_i \omega_i^2 = -\frac{\omega_{\rm I}^2}{z^2} \ . \tag{B16}$$

To leading order we find

$$\left[\kappa_e^2 + F_{\rm I}(z)\right] \ln\!\left\{1 + \frac{F_{\rm I}(z)}{\kappa_e^2}\right\} = -\frac{\omega_{\rm I}^2}{z^2} \ , \tag{B17}$$

and therefore

$$\bar{\mathcal{C}}^0_{e{\rm I},{\rm R}} \ = \ \omega_{\rm I}^2 \lim_{V\to\infty} \frac{i}{2\pi} \int_{C_V} \frac{dz}{z} \bigg|_{z=Ve^{i\theta}} = \omega_{\rm I}^2 \frac{i}{2\pi} \int_0^{\pi} id\theta = -\frac{\omega_{\rm I}^2}{2} \ . \tag{B18}$$

To verify (B16) note the following:

$$F_{\rm I}(z) = -\sum \kappa_i^2 \left(\frac{\beta_i m_i}{2\pi}\right)^{1/2} \int_{-\infty}^{\infty} du\, \frac{u}{z}\left(1-\frac{u}{z}-i\eta\right)^{-1} e^{-\frac{1}{2}\beta_i m_i u^2} \tag{B19}$$

$$= -\sum \kappa_i^2 \left(\frac{\beta_i m_i}{2\pi}\right)^{1/2} \int_{-\infty}^{\infty} du\, \frac{u}{z}\left(1+\frac{u}{z}+\cdots\right) e^{-\frac{1}{2}\beta_i m_i u^2} \tag{B20}$$

$$= -\sum \kappa_i^2 \left(\frac{\beta_i m_i}{2\pi}\right)^{1/2} \frac{1}{z^2} \int_{-\infty}^{\infty} du\, u^2 e^{-\frac{1}{2}\beta_i m_i u^2} \tag{B21}$$

$$= -\sum \kappa_i^2 \left(\frac{\beta_i m_i}{2\pi}\right)^{1/2} \frac{1}{z^2}\,\frac{\sqrt{\pi}}{2}\left(\frac{2}{\beta_i m_i}\right)^{3/2} = -\frac{1}{z^2}\sum\nolimits_i \frac{\kappa_i^2}{\beta_i m_i} \tag{B22}$$

$$= -\frac{1}{z^2}\sum\nolimits_i \omega_i^2\ . \tag{B23}$$

Appendix C: F-Function: frfi

We can write the dielectric function (1.9) as a sum over plasma

$$F(v) \;=\; -\sum\nolimits_b \int_{-\infty}^{\infty} du\, \frac{\rho_b(v)}{v-u+i\eta} \tag{C1}$$

$$\rho_b(v) \;=\; \kappa_b^2 \sqrt{\frac{\beta_b m_b}{2\pi}}\, v\, \exp\!\left\{-\frac{1}{2}\,\beta_b m_b\, v^2\right\}\;. \tag{C2}$$

For numerical work it is best to use the explicit real and imaginary parts of F, which can be written

$$F_{\rm Re}(v) \;=\; \sum_b \kappa_b^2 \left[1 - 2\sqrt{\frac{\beta_b m_b}{2}}\; v\; \mathrm{daw}\left\{\sqrt{\frac{\beta_b m_b}{2}}\, v\right\}\right] \tag{C3}$$

$$F_{\rm Im}(v) \;=\; \sqrt{\pi}\sum_b \kappa_b^2 \sqrt{\frac{\beta_b m_b}{2}}\;\; v\, \exp\left\{-\frac{\beta_b m_b}{2}\, v^2\right\} = \pi\rho_{\rm tot}(v)\;, \tag{C4}$$

where the Dawson integral is defined by

$$\mathrm{daw}(x) = \int_0^x dy\, e^{y^2-x^2} = \frac{\sqrt{\pi}}{2}\, e^{-x^2}\mathrm{erfi}(x)\;. \tag{C5}$$

The function F takes the form

$$F(v) = \sum\nolimits_b \mathbb{F}_b\,(a_b) \quad \text{with} \quad a_b = \left(\frac{\beta_b m_b}{2}\right)^{1/2} v = \left(\frac{\beta_b\, \bar{m}_b}{2}\right)^{1/2} \frac{v}{c} \tag{C6}$$

where $\bar{m}_b = m_b c^2$.

rate.f90:frfi

```
SUBROUTINE frfi(x,nni,alfb,ab,fr,fi,fabs,farg)
USE mathvars
IMPLICIT NONE
REAL,                          INTENT(IN)  :: x
INTEGER,                       INTENT(IN)  :: nni    ! Number of ion species
REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: alfb   ! alpha(b)
REAL,    DIMENSION(1:nni+1), INTENT(IN)  :: ab     ! a(b)
REAL,                          INTENT(OUT) :: fr     ! Dimensionless: multiply fr
REAL,                          INTENT(OUT) :: fi     ! fi by ke^2 for physical un
REAL,                          INTENT(OUT) :: fabs   !
REAL,                          INTENT(OUT) :: farg   !

REAL                :: xib, daw
INTEGER             :: ib
fr=0
fi=0
DO ib=1,nni+1
  xib=ab(ib)*x
  fr=fr + alfb(ib)*(1-2*xib*daw(xib))
  fi=fi + alfb(ib)*xib*EXP(-xib*xib)
ENDDO
fi=fi*SQRT(PI)
fabs=SQRT(fr*fr + fi*fi)
farg=ATAN2(fi,fr)
END SUBROUTINE frfi
```